
Solución al Trabajo Práctico - Convocatoria extraordinaria 2025

Ejercicio 1

Se desea diseñar un circuito digital que implemente las funciones X e Y mostradas a

continuación, que dependen de las cuatro variables A, B, C y D:

X = A'B'D' + A'BC' + A'BD + AB'D

Y = A'BD' + AB'D + A'BD

1.a) (0.5 puntos) Escriba en VHDL la entity del circuito que implemente las dos

funciones lógicas. Es decir, que tenga cuatro entradas A, B, C y D, y dos salidas

X e Y.

1.b) (1 punto) Escriba en VHDL la architecture que describa el comportamiento

del circuito.

1.c) (0.5 puntos) Dibuje el diagrama de un circuito que implemente estas dos

funciones lógicas al nivel de puertas lógicas. No es necesario que el circuito esté

simplificado. A continuación, escriba en VHDL la entity y la architecture de

cada una de las puertas lógicas que componen el circuito que acaba de dibujar.

1.d) (1 punto) Escriba en VHDL una architecture que describa la estructura del

circuito que ha dibujado, instanciando y conectando las puertas lógicas que ha

diseñado anteriormente.

1.e) (1 punto) Escriba en VHDL un banco de pruebas que permita visualizar, para

todos los posibles valores de las entradas, las salidas de los circuitos diseñados

en los Apartados 1.b y 1.d. Compruebe mediante inspección visual que los dos

diseños funcionan correctamente.

Solución al Ejercicio 1

El Código VHDL 1.1 y 1.2 es una posible solución a los Apartados 1.a y 1.b.

El diagrama al nivel de puertas lógicas pedido en el Apartado 1.c está representado

en la Figura 1.1.

El Código VHDL 1.3 – 1.5 es una posible descripción de las puertas lógicas AND

de tres entradas, NOT y OR de tres entradas. El Código VHDL 1.6 es una posible

descripción estructural del circuito, que corresponde con el diagrama mostrado en

la Figura 1.1.

1 library IEEE;

2 use IEEE.std_logic_1164.all;

3

4 entity funcLog is

5 port ( X, Y : out std_logic;

6 A, B, C, D : in std_logic );

7 end entity funcLog;

Código VHDL 1.1: Apartado 1.a: entity del circuito.

1 architecture Comp of funcLog is

2 begin

3 X <= (not A and not B and not D ) or

4 (not A and B and not C) or (not A and B and D) or

5 (A and not B and D) ;

6 Y <= (not A and B and not D) or (A and not B and D) or

7 (not A and B and D);

8 end architecture Comp;

Código VHDL 1.2: Apartado 1.b: architecture que describe el comportamiento.

1 library IEEE;

2 use IEEE.std_logic_1164.all;

3

4 entity and3 is

5 port ( y0 : out std_logic;

6 x0, x1, x2 : in std_logic );

7 end entity and3;

8 architecture and3 of and3 is

9 begin

10 y0 <= x0 and x1 and x2;

11 end architecture and3;

Código VHDL 1.3: Apartado 1.c: puerta AND de tres entradas.

1 library IEEE;

2 use IEEE.std_logic_1164.all;

3

4 entity not1 is

5 port ( y0 : out std_logic;

6 x0 : in std_logic );

7 end entity not1;

8 architecture not1 of not1 is

9 begin

10 y0 <= not x0;

11 end architecture not1;

Código VHDL 1.4: Apartado 1.c: puerta NOT.

1 library IEEE;

2 use IEEE.std_logic_1164.all;

3

4 entity or3 is

5 port ( y0 : out std_logic;

6 x0, x1, x2 : in std_logic );

7 end entity or3;

8 architecture or3 of or3 is

9 begin

10 y0 <= x0 or x1 or x2;

11 end architecture or3;

Código VHDL 1.5: Apartado 1.c: puerta OR de tres entradas.

Figura 1.1: Apartado 1.c: diagrama al nivel de puertas lógicas.

1 library IEEE;

2 use IEEE.std_logic_1164.all;

3 architecture circuito_Estruc of funcLog is

4 signal An, Bn,Cn, Dn, sA1: std_logic;

5 signal sA2, sA3, sA4, sA5, sO1: std_logic;

6 -- Declaración de las clases de los componentes

7 component and3 is

8 port ( y0 : out std_logic ;

9 x0, x1, x2 : in std_logic);

10 end component and3;

11 component not1 is

12 port ( y0 : out std_logic;

13 x0 : in std_logic );

14 end component not1;

15 component or3 is

16 port ( y0 : out std_logic;

17 x0, x1, x2 : in std_logic );

18 end component or3;

19 begin

20 -- Instanciación y conexión de los componentes

21 N1 : component not1 port map (An, A);

22 N2 : component not1 port map (Bn, B);

23 N3 : component not1 port map (Cn, C);

24 N4 : component not1 port map (Dn, D);

25 A1 : component and3 port map (sA1, An, Bn, Dn);

26 A2 : component and3 port map (sA2, An, B, Cn);

27 A3 : component and3 port map (sA3, An, B, D);

28 A4 : component and3 port map (sA4, A, Bn, D);

29 A5 : component and3 port map (sA5, An, B, Dn);

30 O1 : component or3 port map (sO1, sA1, sA2, sA3);

31 O2 : component or3 port map (X, sO1, sA4, '0');

32 O3 : component or3 port map (Y, sA5, sA3, sA4);

33 end architecture circuito_Estruc;

Código VHDL 1.6: Apartado 1.d: descripción estructural al nivel de puertas lógicas.

Finalmente, el Código VHDL 1.7 es un banco de pruebas para los diseños del

circuito que implementa las dos funciones lógicas. En las Figuras 1.2–1.3 se muestran

las formas de onda obtenidas al simular el banco de pruebas con los dos diseños

anteriormente realizados. La comprobación de que el diseño funciona correctamente

debe hacerse en este caso mediante inspección visual.

1 library IEEE;

2 use IEEE.std_logic_1164.all;

3 use IEEE.numeric_std.all;

4

5 entity bp_funcLog is

6 constant DELAY : time := 20 ns; -- Retardo usado en el test

7 end entity bp_funcLog;

8

9 architecture bp_funcLog of bp_funcLog is

10 signal X, Y : std_logic;

11 signal A, B, C, D : std_logic;

12

13 component funcLog is

14 port ( X, Y : out std_logic;

15 A, B, C, D : in std_logic );

16 end component funcLog;

17

18 begin

19 UUT : component funcLog port map

20 (X, Y, A, B, C, D);

21

22 vec_test : process is

23 variable valor : unsigned (3 downto 0);

24 begin

25 -- Generar todos los posibles valores de entrada

26 for i in 0 to 15 loop

27 valor := to_unsigned(i,4);

28 A <= std_logic(valor(3));

29 B <= std_logic(valor(2));

30 C <= std_logic(valor(1));

31 D <= std_logic(valor(0));

32 wait for DELAY;

33 end loop;

34 wait; -- Final de la simulación

35 end process vec_test;

36 end architecture bp_funcLog;

Código VHDL 1.7: Apartado 1.e: banco de pruebas.

Figura 1.2: Apartado 1.e: simulación del banco de pruebas del diseño describiendo el comporta-

miento del circuito.

Figura 1.3: Apartado 1.e: simulación del banco de pruebas del diseño describiendo la estructura

del circuito.

Ejercicio 2

Se quiere diseñar un circuito secuencial conversor serie-a-paralelo de 8 bits que opera

en el flanco de subida de la señal de reloj y tiene una entrada de reset. El circuito

tiene la siguiente entity:

entity conversor is

port( DataOut: out std_logic_vector( 7 downto 0);

Valid: out std_logic;

Clk: in std_logic;

Reset: in std_logic;

Start: in std_logic;

DataIn: in std_logic );

end entity conversor;

A continuación se describe el funcionamiento del circuito conversor.

Los bits recibidos por la entrada serie DataIn se almacenan en un registro interno

del circuito llamado Content, tal como se describe a continuación.

En el flanco de subida de la señal de reloj, si la señal Valid vale '0' y la señal

Reset vale '0', se introduce el valor de la señal de entrada serie DataIn en la

posición más significativa del registro Content desplazando el contenido del registro

a la derecha. El contenido del registro Content es reseteado al valor "00000000"

en el flanco de subida de la señal de reloj cuando la señal Reset vale '1'. Si la

señal Reset vale '0' y la señal Valid vale '1', se mantiene el valor del registro

Content.

La señal de salida Valid se pone a '0' cuando la señal de entrada Start vale '1'

en el flanco de subida de la señal de reloj. Si a partir de ese momento se producen 7

flancos consecutivos de subida de la señal de reloj en los que la señal Start valga

'0', la señal Valid pasa a tomar el valor '1' en el séptimo flanco de subida de

la señal de reloj. El valor de la señal Valid se mantiene a '1' siempre que la

señal Start valga '0' en los flancos de subida de la señal de reloj. En el siguiente

cronograma pueden verse los valores que toma la señal de salida Valid en función

de los valores que tienen la señal de reloj Clk y la señal de entrada Start.

Clk

Start

Valid

El contenido del registro Content se carga en la salida paralelo DataOut siempre

que la señal Valid tenga el valor '1' y la señal Reset valga '0'. Si la señal Valid

tiene el valor '0' y la señal Reset vale '0', la señal DataOut mantiene su valor.

La salida DataOut se resetea al valor "00000000" cuando la señal Reset vale

'1'.

2.a) (3 puntos) Diseñe en VHDL la architecture que describe el comportamiento

del circuito secuencial descrito anteriormente.

2.b) (3 puntos) Programe en VHDL el banco de pruebas del circuito que ha

diseñado en el anterior apartado. La señal de reloj (Clk) debe tener un periodo

de 20 ns e inicialmente valer '0'. El primer flanco de subida de la señal de

reloj se ha de producir en el instante 10 ns. El programa de test debe realizar

consecutivamente las acciones siguientes:

Dpto. de Informática y Automática, UNED 7

1. Reset. La señal de reset ha de tener el valor '1' durante los primeros 15

ns.

2. Cargar en la señal de salida paralelo DataOut el valor "11111100".

La señal Start ha de tener el valor '1' únicamente entre los instantes

40 ns y 60 ns, debiendo tener el valor '0' el resto del tiempo. La señal

de entrada DataIn ha de valer inicialmente '0' y cambiar al valor '1'

en el instante 80 ns. En consecuencia, la señal DataIn toma en ciclos

de reloj consecutivos los valores '0', '0', '1', '1', '1', '1', '1' y

'1'. Comprobar que las señales de salida DataOut y Valid toman los

valores correctos. En caso contrario, el banco de pruebas debe mostrar

un mensaje indicándolo.

Solución al Ejercicio 2

El Código VHDL 1.8 muestra el diseño del registro.

1 library IEEE;

2 use IEEE.std_logic_1164.all;

3 use IEEE.std_logic_arith.all;

4 use IEEE.std_logic_unsigned.all;

5

6 entity conversor is

7 port( DataOut: out std_logic_vector( 7 downto 0);

8 Valid: out std_logic;

9 Clk: in std_logic;

10 Reset: in std_logic;

11 Start: in std_logic;

12 DataIn: in std_logic );

13 end entity conversor;

14

15 architecture conversor of conversor is

16 signal Content: std_logic_vector(7 downto 0);

17 signal ValidInt: std_logic;

18 signal Cnt: integer range 0 to 8;

19 begin

20 conv: process(Clk)

21 begin

22 if rising_edge(Clk) then

23 if (Reset = '1') then

24 Content <= (others => '0');

25 elsif (ValidInt = '0') then

26 Content <= DataIn & Content(7 downto 1);

27 end if;

28 end if;

29 end process;

30

31 outP: process(Reset, Content, ValidInt)

32 begin

33 if (Reset = '1') then

34 DataOut <= (others => '0');

35 elsif (ValidInt = '1') then

36 DataOut <= Content;

37 end if;

38 end process;

39

40 cuenta: process(Clk)

41 begin

42 if (rising_edge(Clk)) then

43 if (Start = '1') then

44 Cnt <= 0;

45 elsif (Cnt /= 7) then

46 Cnt <= Cnt + 1;

47 end if;

48 end if;

49 end process;

50

51 validez: process(Cnt)

52 begin

53 ValidInt <= '0';

54 if (Cnt = 7) then

55 ValidInt <= '1';

56 end if;

57 end process;

58 Valid <= ValidInt;

59 end architecture conversor;

Código VHDL 1.8: Apartado 2.a: diseño del circuito conversor.

El banco de pruebas del circuito se muestra en el Código VHDL 1.9. El cronograma

obtenido al simular el banco de pruebas, usando como circuito a testear el diseño

del Apartado 2.a, se muestra en la Figura 1.4.

1 library IEEE;

2 use IEEE.std_logic_1164.all;

3 use IEEE.numeric_std.all;

4 entity bp_conversor is

5 constant PERIODO : time := 20 ns; -- PERIODO clk

6 end entity bp_conversor;

7

8 architecture bp_conversor of bp_conversor is

9 signal DataOut: std_logic_vector(7 downto 0); --Salida UUT

10 signal Valid : std_logic; --Salida UUT

11 signal Clk : std_logic := '0';

12 signal Start: std_logic := '0';

13 signal Reset, DataIn: std_logic;

14

15 component conversor is

16 port( DataOut: out std_logic_vector( 7 downto 0);

17 Valid: out std_logic;

18 Clk: in std_logic;

19 Reset: in std_logic;

20 Start: in std_logic;

21 DataIn: in std_logic );

22 end component conversor;

23

24 begin

25 UUT : component conversor port map

26 (DataOut, Valid, Clk, Reset, Start, DataIn);

27 Clk <= not Clk after (PERIODO/2);

28 reset <= '1' , '0' after 15 ns;

29 vec_test : process is

30 variable temp : std_logic_vector (7 downto 0);

31 begin

32 report "Comienza la simulación";

33 temp := "11111100";

34 wait for 2*PERIODO;

35 start <= '1';

36 for i in 0 to 7 loop

37 DataIn <= temp(i);

38 wait for PERIODO;

39 if (i = 0) then

40 start <= '0';

41 end if;

42 if i<7 then

43 assert(Valid='0') report("Valor de Y erroneo");

44 assert(DataOut = "00000000") report ("Valor de DataOut erroneo

");

45 else

46 assert(Valid='1') report("Valor de Y erroneo");

47 assert(DataOut = temp) report ("Valor de DataOut erroneo");

48 end if;

49 end loop;

50 report("Simulacion finalizada");

51 wait; -- Termina la simulación

52 end process vec_test;

53 end architecture bp_conversor;

Código VHDL 1.9: Apartado 2.b: banco de pruebas del circuito.

Figura 1.4: Cronograma del Apartado 2.b comprobando el comportamiento del circuito conversor.

